

s = one space in a PRINT statement
m = multiplication sign
d = division sign
0 = zero (the letter O is not used)

NOTE: use command words (e.g., LINE) rather than individual letters in PRINT statements whenever possible to save memory space.

MEMORY MAZE (C)1980 BY S L WALTERS

This typed listing has some extra spaces included for easier reading (for example, FOR N=1 TO 5 and IF S=7 S=1). Do not input any spaces which are not needed unless specifically indicated by an "s" in the listing.

```
1 .MAZE
2 GOTO 98
3 PRINT "EASYss",; RETURN
4 NT=10; MU=M; MU=M; NT=40; MU=M; RETURN
5 PRINT "MEDIUM",; RETURN
7 PRINT "HARDss",; RETURN
9 CX=-65; CY=-40; RETURN
10 BOX 0,0,153,71,1; RETURN
12 BOX 0,-40,159,7,2; RETURN
14 NT=5; MU=RND(20)+50; NT=0; @(20)=129; RETURN
16 FOR N=1 TO 300; NEXT N; RETURN
17 FOR N=1 TO 2000; NEXT N; RETURN
20 FOR N=0 TO 16
22 FOR H=@(N=17) TO @(N); BOX Nm9-72,Hm9-36,9,9,3; GOSUB 14; NEXT H
24 NEXT N; RETURN
30 BOX Vm9-72,Wm9-36,7,7,2
32 CX=Xm9-71; CY=Ym9-36; PRINT "*",; RETURN
34 BOX 0,Dd2m9-27,153,Dm9,2; RETURN
41 PRINT "BUMP!"; RETURN
42 PRINT "THUD!"; RETURN
43 PRINT "OUCH!"; RETURN
45 GOSUB 9; PRINT #1, "SCORE:s#1=",@(34),
46 IF T=2 PRINT #1, "sss#2=",@(35),
47 RETURN
50 GOSUB 9; PRINT #1, "POINTS:s",S,"sssPEEK-TR",; RETURN
58 X=0; Y=@(17); GOSUB 32; RETURN
70 M=53; GOSUB 4; M=51; GOSUB 4; M=53; GOSUB 4; NT=20; MU=52; NT=60; MU=50; MU=50;
  MU=51; NT=80; MU=49; NT=0; RETURN
80 Z=20; GOSUB 83
81 Z=60; GOSUB 83
82 NT=15; MU=54; MU=52; MU=54; NT=60; MU=56; NT=0; RETURN
83 NT=Z; MU=53; NT=10; MU=53; MU=53; RETURN
98 BOX -31,-32,99,7,2; IF TR(1)#1 GOTO 2
99 CLEAR
100 CX=-31; PRINT "MEMORY MAZE"; GOSUB 10; D=7; GOSUB 900; GOSUB 20; GOSUB 58; GOSUB 32
140 GOSUB 9; PRINT "DIAL DIFFICULTY:s",
150 D=KN(1)d80m2+5; GOSUB D; IF TR(1)#1 GOTO 140
180 GOSUB 12; GOSUB 16
200 GOSUB 9; PRINT "NO. OF PLAYERS:s",
210 K=KN(1); T=1; IF K>0 T=2
220 PRINT #1,T,
230 IF TR(1)#1 GOTO 200
240 GOSUB 12; GOSUB 16; @(34)=0; @(35)=0; P=0
300 GOSUB 10; P=P+1; IF P>T P=1
310 GOSUB 9; PRINT #1, "#",P,"sSTUDY:s10",; GOSUB 900; GOSUB 20; GOSUB 58
370 FOR N=1 TO 1000; Z=Nd100; IF RM=0 CX=-5; CY=-40; PRINT #2, 10-Z,
376 NEXT N; GOSUB 12; GOSUB 34; S=9; GOSUB 50; GOSUB 32
400 V=X; W=Y; X=X+JX(P); IF X<0 X=0
414 IF X>16 X=16
420 Y=Y+JY(P); IF Y<1 Y=1
424 IF Y>D Y=D
430 IF (Y<@(X+17))+(Y>@(X)) GOTO 700
440 GOSUB 30; IF X=16 GOTO 500
460 IF TR(P) GOTO 800
```

```

470 GOSUB 16; GOTO 400
500 GOSUB 12; GOSUB 9; IF S<0 PRINT "YOU'RE LOST!"; GOSUB 70; GOTO 530
520 PRINT "YOU MADE IT!"; GOSUB 80
530 GOSUB 17; @(P+33)=@(P+33)+S; GOSUB 12; GOSUB 9; PRINT "INSTANT REPLAY"; GOSUB 10;
    GOSUB 20; GOSUB 32; GOSUB 17; GOSUB 12; GOSUB 9; GOSUB 45; GOSUB 17; GOSUB 12; GOTO 300
700 CX=CX-24; IF CX<-72 CX=-72
710 GOSUB RND(3)+40; BC=101; &(20)=10; GOSUB 14; X=V; Y=W; S=S-1; GOSUB 50; GOSUB 16;
    GOSUB 16; BC=133; IF S<0 GOTO 500
750 GOSUB 34; GOSUB 32; GOTO 400
800 GOSUB 12; GOSUB 9; PRINT "TAKE A PEEK!",
820 S=S-3; IF S<0 GOTO 500
830 GOSUB 10; GOSUB 20; GOSUB 32; GOSUB 17; GOSUB 34; GOSUB 32; GOSUB 50; GOTO 440
900 @(0)=RND(D); @(17)=@(0)
910 FOR N=1 TO 15 STEP 2; GOSUB 14; A=D-@(N-1)+1; @(N)=RND(A)+@(N-1)-1; B=@(N-1);
    @(N+17)=RND(B); C=@(N)-@(N+17)+1
920 @(N+1)=RND(C)-1+@(N+17); @(N+18)=@(N+1); NEXT N; RETURN

```

November 28, 1980

The program must be loaded to tape with the following procedure to set certain variables required by the operating program:

```

:PRINT; TV=0; TV=7; PRINT; PRINT ".MEMORY MAZE"; PRINT ".(C)1980 BY S L WALTERS";
PRINT; LIST; PRINT; PRINT ".STANDBY FOR"; PRINT ":RETURN; :INPUT 8"

```

When the last part appears on the screen, stop the tape but do not rewind it. Then enter the following:

```

:PRINT; TV=0; TV=8; PRINT "&(9)=169; &(20)=129; BC=133; :RETURN; NT=0; RUN"

```

Start the tape on record again and push GO. When this is finished turn the tape off, and it is ready to be reloaded when needed.

The above can be done without clearing the program from the computer memory, so you can verify the results with :LIST before resetting the computer if you want to.

OPERATING THE PROGRAM

The program starts with selection of 1 or 2 players, and then the level of difficulty - EASY, MEDIUM or HARD. The maze shown while the difficulty level is being selected is a HARD maze.

Pull the trigger to start the game, and a new maze appears for you to study while the computer counts down from 10 to 0 (about 30 seconds). When this is done, the maze becomes invisible.

Try to move the marker through the maze without bumping into any walls. If you do, the program lets you know, and subtracts one point from you potential score.

If you get completely lost, and have at least 3 points left, you can pull the trigger and take a peek at the maze. It will cost you 3 points to do this.

If you bump into a wall when you have no points left (or try to take a peek with less than 3 points left) the computer will decide you are hopelessly lost and end the game.

An instant replay feature then shows you the maze and where you were in it, and shows you your total score for all games played. After a brief delay, the computer then proceeds automatically to the next game.